


## REALM FACET

Moving down a little deeper, one eventually finds the Gotchiverse Realm, which is located beyond the Human Realm, it's a place where smart contracts glitter. The ghost of a yield farmer who is wiped off in the Ether Realm travels to the Gotchiverse to become an Aavegotchi. The Aavegotchi are a unique species who like nothing more than farming and voting. With curiosity human want to know how the Realm was established which led human to checking the source of the Realm (RealmFacet.sol)


## License and Compiler



```
1 // SPDX-License-Identifier: UNLICENSED
2 pragma solidity 0.8.9;
```

The license Identifier used in the RealmFacet.sol is unlicensed which means anyone is free to copy, modify, publish, use, compile, sell, or distribute this software, either in source code form or as a compiled binary, for any purpose, commercial or non-commercial, and by any means.

The Solidity version used in this contract is **0.8.9**, this require a compiler version of 0.8.9.



```

1  import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
2  import "../../libraries/AppStorage.sol";
3  import "../../libraries/LibDiamond.sol";
4  import "../../libraries/LibStrings.sol";
5  import "../../libraries/LibMeta.sol";
6  import "../../libraries/LibERC721.sol";
7  import "../../libraries/LibRealm.sol";
8  import "../../libraries/LibAlchemica.sol";
9  import {InstallationDiamondInterface} from "../../interfaces/InstallationDiamondInterface.sol";
10 import "../../libraries/LibSignature.sol";
11 import "../../interfaces/IERC1155Marketplace.sol";


```

### 1. **import "@open-zeppelin/contracts/token/ERC20/IERC20.sol";**

The first import in this , it is an interface for ERC20, the interface gives a framework for the EIP20 standard.

### 2. **import "../../libraries/AppStorage.sol";**

The line two is a library import from AppStorage.sol. It is important to note that this contract is an upgradeable contract using the diamond standard for upgradability, the AppStorage.sol has some struct in it which act as a layout for state variables to be used by RealmFacet.sol, the library named LibAppStorage has a function in it pointing to slot zero which can be used within a contract or library function.



```

1  library LibAppStorage {
2      function diamondStorage() internal pure returns (AppStorage storage ds) {
3          assembly {
4              ds.slot := 0
5          }
6      }
7  }

```

Also in the AppStorage.sol is a contract only for modifiers which is inherited by the RealmFacet.sol and contains the usage of the state variable layout declared in slot 0 of the **contract Modifiers**.

```
1  contract Modifiers {
2      AppStorage internal s;
3
4      modifier onlyParcelOwner(uint256 _tokenId) {
5          require(LibMeta.msgSender() == s.parcelOwner[_tokenId], "AppStorage: Only Parcel owner can call");
6          _;
7      }
8
9      modifier onlyOwner() {
10         LibDiamond.enforceIsContractOwner();
11         _;
12     }
13
14     modifier onlyGotchiOwner(uint256 _gotchiId) {
15         AavegotchiDiamond diamond = AavegotchiDiamond(s.aavegotchiDiamond);
16         require(LibMeta.msgSender() == diamond.ownerOf(_gotchiId), "AppStorage: Only Gotchi Owner can call");
17         _;
18     }
19
20     modifier onlyGameManager() {
21         require(msg.sender == s.gameManager, "AlchemicaFacet: Only Game Manager");
22         _;
23     }
24
25     modifier onlyInstallationDiamond() {
26         require(LibMeta.msgSender() == s.installationsDiamond, "AppStorage: Only Installation diamond can call");
27         _;
28     }
29
30     modifier gameActive() {
31         require(s.gameActive, "AppStorage: game not active");
32         _;
33     }
34
35     modifier canBuild() {
36         require(!s.freezeBuilding, "AppStorage: Building temporarily disabled");
37         _;
38     }
39 }
```

### 3. import "../libraries/LibDiamond.sol";

A LibDiamond.sol library import is a default library that helps in the implementation of EIP2535 standard

### 4. import "../libraries/LibStrings.sol";

This is a library import that uses open Zeppelin contract for converting decimals to ASCII string representation.

### 5. import "../libraries/LibMeta.sol";

The LibMeta.sol library is helps in getting relayed call sender, a kind of call origin.

**6. import "../libraries/LibERC721.sol";**

This is a library that implement some of the functions of the EIP721 standard

**7. import "../libraries/LibRealm.sol";**

The LibRealm.sol is a library that helps in some functions used in the build up of the RealmFacet.sol which will later be implemented in some reviews to come.

**8. import "../libraries/LibAlchemica.sol";**

This library importation is responsible for Gotchus Alchemica which are tokens that can be obtained in the Gotchiverse. Four tokens that can be obtained in the Gotchiverse are FUD, KUK, ALPHA, FOMO, these tokens can be used to buy installations like building of wall in the Gotchiverse or sold.

**9. import {InstallationDiamondInterface} from  
"../interfaces/InstallationDiamondInterface.sol";**

An interface for installationDiamond.sol. Remember, installation in Gotchiverse are alter, harvester, reservoir, gotchi lodge, wall, NFT display, buildqueue booster

**10. import "../libraries/LibSignature.sol";**

A library responsible for getting if a user is who he/she claimed to be, it is validated by telling a user to sign a hashed message.

**11. import "../interfaces/IERC1155Marketplace.sol";**

An interface for ERC1155 Market Place, this interface only has one function updateERC1155Listing.

This contract to be reviewed inherit contract Modifiers for its modifiers declaration.



```
1  contract RealmFacet is Modifiers
```

## EVENTS

```
1 event EquipInstallation(uint256 _realmId, uint256 _installationId, uint256 _x, uint256 _y);
2 event UnequipInstallation(uint256 _realmId, uint256 _installationId, uint256 _x, uint256 _y);
3 event EquipTile(uint256 _realmId, uint256 _tileId, uint256 _x, uint256 _y);
4 event UnequipTile(uint256 _realmId, uint256 _tileId, uint256 _x, uint256 _y);
5 event AavegotchiDiamondUpdated(address _aavegotchiDiamond);
6 event InstallationUpgraded(uint256 _realmId, uint256 _prevInstallationId, uint256 _nextInstallationId, uint256 _coordinateX, uint256 _coordinateY);
7
```

Event **EquipInstallation** will be emitted whenever an Aavegotchi equip an installation. The parameter `realmId`, the `installationId` of the user's purchase and the coordinate `x` and `y` will be emitted, whenever an Aavegotchi unequipped an installation the **UnequipInstallation** event will be emitted. For the event **EquipTile**, the event will be logged when an Aavegotchi equipped a tile. Tiles are craft-able cosmetics also known as aesthetica, they can be placed on Gotchiverse land. Event `unequipped` is being logged out whenever a user unequipped a tile. The event **AavegotchiDiamondUpdated** is being emitted whenever there's an update to the AavegotchiDiamond. Lastly, the event **InstallationUpgraded** is being emitted whenever there's an upgrade to the installation, the upgrade in this case is like acquiring a higher installation compared to the previous, it takes in parameter `realmId`, the previous `installationId`, the next `installationId`, the coordinate `x` and `y` all in `uint256`.

## FUNCTIONS EXPLANATION

Starting with the function mint parcels **Function mintParcels**

```
1 function mintParcels(  
2   address[] calldata _to,  
3   uint256[] calldata _tokenIds,  
4   MintParcelInput[] memory _metadata  
5 ) external onlyOwner {  
6   for (uint256 index = 0; index < _tokenIds.length; index++) {  
7     require(s.tokenIds.length < LibRealm.MAX_SUPPLY, "RealmFacet: Cannot mint more than 420,069 parcels");  
8     uint256 tokenId = _tokenIds[index];  
9     address toAddress = _to[index];  
10    MintParcelInput memory metadata = _metadata[index];  
11    require(_tokenIds.length == _metadata.length, "Inputs must be same length");  
12    require(_to.length == _tokenIds.length, "Inputs must be same length");  
13  
14    Parcel storage parcel = s.parcels[tokenId];  
15    parcel.coordinateX = metadata.coordinateX;  
16    parcel.coordinateY = metadata.coordinateY;  
17    parcel.parcelId = metadata.parcelId;  
18    parcel.size = metadata.size;  
19    parcel.district = metadata.district;  
20    parcel.parcelAddress = metadata.parcelAddress;  
21    parcel.alchemicaBoost = metadata.boost;  
22  
23    LibERC721.safeMint(toAddress, tokenId);  
24  }  
25 }
```

```
1 struct MintParcelInput {  
2   uint256 coordinateX;  
3   uint256 coordinateY;  
4   uint256 district;  
5   string parcelId;  
6   string parcelAddress;  
7   uint256 size; //0=humble, 1=reasonable, 2=spacious vertical, 3=spacious horizontal, 4=partner  
8   uint256[4] boost; //fud, fomo, alpha, kek  
9 }
```

This function is an external function with an `onlyOwner` modifier that has been declared in the **contract modifier**, which means it can only be called externally by the diamond owner (restricted to the diamond owner). It uses the `LibDiamond.enforceIsContractOwner` to ensure the function is only called by the diamond owner.

The function allows the diamond owner to mint new parcels. It takes In 3 parameters, the array of addresses to mint to, the tokenIds of token (parcel) to mint and an array of structs containing the metadata of each parcel being minted. The struct contains the parcel coordinate both in x and y axis, the district in number, the parcelId, the parcelAddress, the size which can be either (humble, reasonable, spacious vertical, spacious horizontal or partner). Lastly, the boost which is an array of the gotchus alchemica. Parcel in Gotchiverse are piece of land which are categorized into humble, reasonable, spacious vertical, spacious horizontal and partner, the type of parcel determines the quantity of gotchus alchemica that will be minted on the parcel. Each parcel is represented as an ERC721 NFT and the tokenId of each parcel is used as a pointer to **struct Parcel** in **AppStorage.sol** which is used to store each parcel metadata before safe minting.

## Function batchEquip

```
1 function batchEquip(
2     uint256 _realmId,
3     uint256 _gotchiId,
4     BatchEquipIO memory _params,
5     bytes[] memory _signatures
6 ) external gameActive canBuild {
7     require(_params.ids.length == _params.x.length, "RealmFacet: Wrong length");
8     require(_params.x.length == _params.y.length, "RealmFacet: Wrong length");
9
10    for (uint256 i = 0; i < _params.ids.length; i++) {
11        if (_params.types[i] == 0 && _params.equip[i]) {
12            equipInstallation(_realmId, _gotchiId, _params.ids[i], _params.x[i], _params.y[i], _signatures[i]);
13        } else if (_params.types[i] == 1 && _params.equip[i]) {
14            equipTile(_realmId, _gotchiId, _params.ids[i], _params.x[i], _params.y[i], _signatures[i]);
15        } else if (_params.types[i] == 0 && !_params.equip[i]) {
16            unequipInstallation(_realmId, _gotchiId, _params.ids[i], _params.x[i], _params.y[i], _signatures[i]);
17        } else if (_params.types[i] == 1 && !_params.equip[i]) {
18            unequipTile(_realmId, _gotchiId, _params.ids[i], _params.x[i], _params.y[i], _signatures[i]);
19        }
20    }
21 }
```



```
1 struct BatchEquipIO {  
2     uint256[] types; //0 for installation, 1 for tile  
3     bool[] equip; //true for equip, false for unequip  
4     uint256[] ids;  
5     uint256[] x;  
6     uint256[] y;  
7 }
```

The function `batchEquip` is an external function with modifier **gameActive** and **canBuild**. These modifiers ensure the game is active (**s.gameActive must be true**) and can build (**s.freezeBuilding must be false**) following the modifier declaration in **contract Modifiers** before any batch equip of installation or tile. The function parameters takes the `realmId` in `uint256`, the `gotchiId` in `uint256` and the `_params` in struct data type. The struct contains array of `uint256` for `types` which can be 0 for installation or 1 for tile, also in the struct is an array of `bool` for variable `equip`, which can be true for equip or false for unequip, the array of `uint256` for `ids`, the array of `uint256` for coordinate `x` and the array of `uint256` for coordinate `y`.

Basically, the function is used for batch equip of installation and tile, in the logic of this function is a call to another function depending on the condition each parameter meet, the function calls are **equipInstallation**, **equipTile**, **unequipInstallation**, and **unequipTile**.



## Function equipInstallation

```
1 function equipInstallation(  
2   uint256 _realmId,  
3   uint256 _gotchiId,  
4   uint256 _installationId,  
5   uint256 _x,  
6   uint256 _y,  
7   bytes memory _signature  
8 ) public gameActive canBuild {  
9   //2 - Equip Installations  
10  LibRealm.verifyAccessRight(_realmId, _gotchiId, 2, LibMeta.msgSender());  
11  require(  
12    LibSignature.isValid(keccak256(abi.encodePacked(_realmId, _gotchiId, _installationId, _x, _y)), _signature, s.backendPubKey),  
13    "RealmFacet: Invalid signature"  
14  );  
15  
16  InstallationDiamondInterface.InstallationType memory installation = InstallationDiamondInterface(s.installationsDiamond).getInstallationType(  
17    _installationId  
18  );  
19  
20  require(installation.level == 1, "RealmFacet: Can only equip lvl 1");  
21  
22  if (installation.installationType == 1 || installation.installationType == 2) {  
23    require(s.parcels[_realmId].currentRound >= 1, "RealmFacet: Must survey before equipping");  
24  }  
25  if (installation.installationType == 3) {  
26    require(s.parcels[_realmId].lodgeId == 0, "RealmFacet: Lodge already equipped");  
27    s.parcels[_realmId].lodgeId = _installationId;  
28  }  
29  if (installation.installationType == 6)  
30    require(s.parcels[_realmId].upgradeQueueCapacity == 1, "RealmFacet: Maker already equipped or altar not equipped");  
31  
32  LibRealm.placeInstallation(_realmId, _installationId, _x, _y);  
33  InstallationDiamondInterface(s.installationsDiamond).equipInstallation(msg.sender, _realmId, _installationId);  
34  
35  LibAlchemica.increaseTraits(_realmId, _installationId, false);  
36  
37  IERC1155Marketplace(s.aavegotchiDiamond).updateERC1155Listing(s.installationsDiamond, _installationId, msg.sender);  
38  
39  emit EquipInstallation(_realmId, _installationId, _x, _y);  
40 }
```

This function allows a parcel owner to equip installation, and it is important to note that installations in Gotchiverse are things built on a parcel like wall, alter, harvester, reservoir, gotchi lodge, NFT display, buildqueue booster. The function is a public function and takes in the same modifiers as **function batchEquip** which are the gameActive modifier and canBuild Modifier. Before an installation is equipped, the access right of a user is being checked using **function verifyAccessRight** from **library LibRealm** which uses a 2D mapping to check for user's access right with parcelId and actions as the key for the 2D mapping, Before equipping installation the LibSignature.isValid requirement must be passed. the LibSignature.isValid is used to validate that users are who they claimed to be by telling a user to sign an hashed message with their private key, in this manner real users can be verified from their signature. We also ensure the level of our installation gotten from the query of getInstallationType in **contract installationFacet** is equal to 1. if all of this requirement are met, the installation will be equipped depending on the condition passed. And the **event EquipInstallation** will be emitted.

## Function unequipInstallation

```
1 function unequipInstallation(
2   uint256 _realmId,
3   uint256 _gotchiId, //will be used soon
4   uint256 _installationId,
5   uint256 _x,
6   uint256 _y,
7   bytes memory _signature
8 ) public onlyParcelOwner(_realmId) gameActive canBuild {
9   require(
10    LibSignature.isValid(keccak256(abi.encodePacked(_realmId, _gotchiId, _installationId, _x, _y)), _signature, s.backendPubKey),
11    "RealmFacet: Invalid signature"
12  );
13
14  //TODO: Prevent unequipping if an upgrade is active for this installationId on the parcel
15
16  InstallationDiamondInterface installationsDiamond = InstallationDiamondInterface(s.installationsDiamond);
17  InstallationDiamondInterface.InstallationType memory installation = installationsDiamond.getInstallationType(_installationId);
18
19  require(!LibRealm.installationInUpgradeQueue(_realmId, _installationId, _x, _y), "RealmFacet: Can't unequip installation in upgrade queue");
20  require(
21    installation.installationType != 0 || s.parcels[_realmId].upgradeQueueCapacity == 1,
22    "RealmFacet: Cannot unequip altar when there is a maker"
23  );
24
25  LibRealm.removeInstallation(_realmId, _installationId, _x, _y);
26  InstallationDiamondInterface(s.installationsDiamond).unequipInstallation(msg.sender, _realmId, _installationId);
27  LibAlchemica.reduceTraits(_realmId, _installationId, false);
28
29  //Process refund
30  if (installationsDiamond.getInstallationUnequipType(_installationId) == 0) {
31    //Loop through each level of the installation.
32    //TODO: For now we can use the ID order to get the cost of previous upgrades. But in the future we'll need to add some data redundancy.
33    uint256 currentLevel = installation.level;
34    uint256[] memory alchemicaRefund = new uint256[](4);
35    for (uint256 index = 0; index < currentLevel; index++) {
36      InstallationDiamondInterface.InstallationType memory prevInstallation = installationsDiamond.getInstallationType(_installationId - index);
37
38      //Loop through each Alchemica cost
39      for (uint256 i; i < prevInstallation.alchemicaCost.length; i++) {
40        //Only half of the cost is refunded
41        alchemicaRefund[i] += prevInstallation.alchemicaCost[i] / 2;
42      }
43    }
44
45    for (uint256 j = 0; j < alchemicaRefund.length; j++) {
46      //don't send 0 refunds
47      if (alchemicaRefund[j] > 0) {
48        IERC20 alchemica = IERC20(s.alchemicaAddresses[j]);
49        alchemica.transfer(msg.sender, alchemicaRefund[j]);
50      }
51    }
52  }
53
54  emit UnequipInstallation(_realmId, _installationId, _x, _y);
55 }
```

This function allows a parcel owner to equip an installation, it has six parameters, `realmId`, `gotchiId`, `installationId`, the `x` and `y` coordinate and signature. It is a public function and has the **modifiers** `onlyParcelOwner(realmId)`, `gameActive`, `canBuild`, the `onlyParcelOwner(realmId)` checks if the caller has parcel. The **function isValid** as explained above is required to be true, a check if an installation is not in upgrade queue must also be passed before an installation is removed. Whenever an installation is being unequipped, it is burnt and half of the gotchus alchemica used in the purchase of the installation will be refunded. When an installation is unequipped the **event UnequipInstallation** is being emitted whenever an installation is unequipped.

## Function moveInstallation

```
1 function moveInstallation(
2   uint256 _realmId,
3   uint256 _installationId,
4   uint256 _x0,
5   uint256 _y0,
6   uint256 _x1,
7   uint256 _y1
8 ) external onlyParcelOwner(_realmId) gameActive canBuild {
9   //Check if upgrade is in progress
10  InstallationDiamondInterface installation = InstallationDiamondInterface(s.installationsDiamond);
11
12  require(installation.parcelInstallationUpgrading(_realmId, _installationId, _x0, _y0) == false, "RealmFacet: Installation is upgrading");
13
14  LibRealm.removeInstallation(_realmId, _installationId, _x0, _y0);
15  emit UnequipInstallation(_realmId, _installationId, _x0, _y0);
16  LibRealm.placeInstallation(_realmId, _installationId, _x1, _y1);
17  emit EquipInstallation(_realmId, _installationId, _x1, _y1);
18 }
```

Function `moveInstallation` is similar to both using `unequipInstallation` and `equipInstallation` at the same time but with changes to the coordinate. The parameters that comes with function `moveInstallation` are the `realmId` which is the identifier of the parcel which the installation is being moved on, `installationId` which is the identifier of the installation being moved, initial `x` and `y` coordinate ( the current `x` and `y` coordinate) and the coordinate a user intend to move to. It has the same modifiers as it is in **function `unequipInstallation`**. The **event `UnequipInstallation`** and **event `EquipInstallation`** are being emitted when an installation is moved from one coordinate to another coordinate.

## Function equipTile

```
1 function equipTile(
2   uint256 _realmId,
3   uint256 _gotchiId,
4   uint256 _tileId,
5   uint256 _x,
6   uint256 _y,
7   bytes memory _signature
8 ) public gameActive canBuild {
9   //3 - Equip Tile
10  LibRealm.verifyAccessRight(_realmId, _gotchiId, 3, LibMeta.msgSender());
11  require(
12    LibSignature.isValid(keccak256(abi.encodePacked(_realmId, _gotchiId, _tileId, _x, _y)), _signature, s.backendPubKey),
13    "RealmFacet: Invalid signature"
14  );
15  LibRealm.placeTile(_realmId, _tileId, _x, _y);
16  TileDiamondInterface(s.tileDiamond).equipTile(msg.sender, _realmId, _tileId);
17
18  IERC1155Marketplace(s.aavegotchiDiamond).updateERC1155Listing(s.tileDiamond, _tileId, msg.sender);
19
20  emit EquipTile(_realmId, _tileId, _x, _y);
21 }
```

The equip tile allows a parcel owner to equip a tile, the parameters are realmId (which is the identifier of the parcel to which the tile is being equipped on), gotchiId, the tileId (which is the identifier of the tile being equipped, the x and y coordinate denote the starting coordinate of the tile and are used to make sure that slot is available on a parcel). The logic in **function equipTile** is similar to that of **function equipInstallation**, they have the same modifiers gameActive and canBuild, it verifies user's access right as equipInstallation. User's signature is also being validated with the address they claim to be using. After passing through these requirements, a Tile can be equipped, ERC1155Listings is updated and **event EquipTile** is being emitted or logged to the frontend.


## Function unequipTile

```
1 function unequipTile(  
2     uint256 _realmId,  
3     uint256 _gotchiId,  
4     uint256 _tileId,  
5     uint256 _x,  
6     uint256 _y,  
7     bytes memory _signature  
8 ) public onlyParcelOwner(_realmId) gameActive canBuild {  
9     require(  
10         LibSignature.isValid(keccak256(abi.encodePacked(_realmId, _gotchiId, _tileId, _x, _y)), _signature, s.backendPubKey),  
11         "RealmFacet: Invalid signature"  
12     );  
13     LibRealm.removeTile(_realmId, _tileId, _x, _y);  
14  
15     TileDiamondInterface(s.tileDiamond).unequipTile(msg.sender, _realmId, _tileId);  
16  
17     emit UnequipTile(_realmId, _tileId, _x, _y);  
18 }
```

function unequipTile allow a parcel owner to unequip a tile, it has six parameters which are the realmId (identifier of the parcel which the tile is being equipped from), gotchiId, tileId (identifier of the tile being unequipped), the x coordinate of a tile, the y coordinate of a tile all in uint256, and signature in bytes (expected to be 65 bytes). All the parameters excluding the signature are being concatenated and hashed together to form a hashed message.

The unequipTile has the same modifiers **onlyParcelOwner(realmId)**, **gameActive** and **canBuild** as that of unequipInstallation, after user claimed they are the real owner of an address a tile will then be unequipped and the unequipTile will be logged. The unequipTile log contains the realmId, tileId, and coordiante x and y.

## function moveTile



```
1  function moveTile(  
2      uint256 _realmId,  
3      uint256 _tileId,  
4      uint256 _x0,  
5      uint256 _y0,  
6      uint256 _x1,  
7      uint256 _y1  
8  ) external onlyParcelOwner(_realmId) gameActive canBuild {  
9      LibRealm.removeTile(_realmId, _tileId, _x0, _y0);  
10     emit UnequipTile(_realmId, _tileId, _x0, _y0);  
11     LibRealm.placeTile(_realmId, _tileId, _x1, _y1);  
12     emit EquipTile(_realmId, _tileId, _x1, _y1);  
13 }
```

The function `moveTile` allows a parcel owner to move a tile from one coordinate to another coordinate, the move is done by removing the parcel owner's initial coordinate and adding a new coordinate for both x and y. The `moveTile` is an external function with modifiers **`onlyParcelOwner(_realmId)`**, **`gameActive`** and **`canBuild`**, once this modifiers are passed users will be able to get a new coordinate for there tile after this is done two event is being emitted or logged. The event `UnequipTile` (having the coordinate x and y of the tile, the `realmId` and the `tileId`) and `EquipTile` (having the coordinate x and y parcel owner wants to move there tile to, the `realmId` and the `tileId`). Function `moveTile` is also similar to move installation in logic and modifiers.

## Function upgradeInstallation

```
1 function upgradeInstallation(  
2     uint256 _realmId,  
3     uint256 _prevInstallationId,  
4     uint256 _nextInstallationId,  
5     uint256 _coordinateX,  
6     uint256 _coordinateY  
7 ) external onlyInstallationDiamond {  
8     LibRealm.removeInstallation(_realmId, _prevInstallationId, _coordinateX, _coordinateY);  
9     LibRealm.placeInstallation(_realmId, _nextInstallationId, _coordinateX, _coordinateY);  
10    LibAlchemica.reduceTraits(_realmId, _prevInstallationId, true);  
11    LibAlchemica.increaseTraits(_realmId, _nextInstallationId, true);  
12    emit InstallationUpgraded(_realmId, _prevInstallationId, _nextInstallationId, _coordinateX, _coordinateY);  
13 }
```

This function is responsible for upgrade to an installation equipped by parcel owner, the function is an external function and has a **modifier onlyInstallationDiamond** which means only installation diamond can call. Function upgradeInstallation has five parameters, \_relamId, \_prevInstallationId (previous installation Id), \_nextInstallationId (next installation ID) and the coordinate x and y. this function is basically changing the installation Id from one to another and event installationUpgraded is logged or emitted.

## Function addUpgradeQueueLength

```
1 function addUpgradeQueueLength(uint256 _realmId) external onlyInstallationDiamond {  
2     s.parcels[_realmId].upgradeQueueLength++;  
3 }
```

function addUpgradeQueueLength increases the length of upgrade queue length for a parcel, it is an external function and has a modifier onlyInstallationDiamond as we have in upgradeInstallation.


## Function subUpgradeQueueLength

```
1 function subUpgradeQueueLength(uint256 _realmId) external onlyInstallationDiamond {  
2     s.parcels[_realmId].upgradeQueueLength--;  
3 }
```



function subUpgradeQueueLength is an inverse of addUpgradeQueueLength, doing the same thing in an opposite direction with an external functional visibility (can only be called externally alone) and onlyInstallationDiamond modifier.


### Function buildingFrozen



```
1 function buildingFrozen() external view returns (bool) {  
2     return s.freezeBuilding;  
3 }
```

Function buildingFrozen returns true or false for s.freezeBuilding, which means s.freezeBuilding has a boolean data type.

### Function setFreezeBuilding



```
1 function setFreezeBuilding(bool _freezeBuilding) external onlyOwner {  
2     s.freezeBuilding = _freezeBuilding;  
3 }
```

This function is responsible for setting freeze building to either true or false, it can only be called by the diamond owner.